UNITED STATES DEPARTMENT OF THE INTERIOR
GEOLOGICAL SURVEY

In Situ Geomechanics of Crystalline and Sedimentary Rocks

Part V: RVT--A FORTRAN Program for an Exact Elastic Solution for Tectonic and Gravity Stresses in Isolated Symmetric Ridges and Valleys

By

William Z. Savage,[1] Philip S. Powers,[1] and Henri S. Swolfs[1]

Open-File Report 84-827

1984

[1]U.S. Geological Survey
Denver, Colorado

IN SITU GEOMECHANICS OF CRYSTALLINE AND SEDIMENTARY ROCKS

PART V: RVT--A FORTRAN PROGRAM FOR AN EXACT ELASTIC SOLUTION FOR TECTONIC AND GRAVITY STRESSES IN ISOLATED SYMMETRIC RIDGES AND VALLEYS

By William Z. Savage, Philip S. Powers, and Henri S. Swolfs

PREFACE

This report is the fifth of a series summarizing the results of the U.S. Geological Survey's research program in geomechanics aimed at investigating and assessing the potential of crystalline and sedimentary rock masses as geologic repositories of nuclear waste. The first four parts of this series of reports are referenced below:

Savage, W. Z., and Swolfs, H. S., 1980, The long-term deformation and time-temperature correspondence of viscoelastic rock--an alternative theoretical approach, Pt. 1 of In situ geomechanics of crystalline and sedimentary rocks: U.S. Geological Survey Open-File Report 80-708, 21 p.

Smith, W. K., 1982, Two BASIC computer programs for the determination of in situ stresses using the CSIRO hollow inclusion stress cell and the USBM borehole deformation gage [Pt. 2 of In situ geomechanics of crystalline and sedimentary rocks]: U.S. Geological Survey Open-File Report 82-489, 40 p.

Swolfs, H. S., 1982, First experiences with the C.S.I.R.O. hollow-inclusion stress cell, Pt. 3 of In situ geomechanics of crystalline and sedimentary rocks: U.S. Geological Survey Open-File Report 82-990, 20 p.

Nichols, T. C., 1983, Continued field testing of the modified U.S. Geololgical Survey 3-D borehole stress probe, Pt. 4 of In situ geomechanics of crystalline and sedimentary rocks: U.S. Geological Survey Open-File Report 83-750, 11 p.

Published journal articles that report on the findings of this program are referenced below:

Swolfs, H. S., and Kibler, J. D., 1982, A note on the Goodman Jack: Rock Mechanics, v. 15, no. 2, p. 57-66.

Swolfs, H. S., 1983, Aspects of the size-strength relationship of unjointed rocks: Chapter 51 in Rock mechanics--Theory-experimenmt-practice: 24th U.S. Symposium on Rock Mechanics, College Station, Texas, p. 501-510.

Swolfs, H. S., 1984, The triangular stress diagram--a graphical representation of crustal stress measurements: U.S. Geological Survey Professional Paper 1291, 19 p.

Swolfs, H. S., and Savage, W. Z, 1984, Site characterization studies of a volcanic cap rock: Chapter 39 in Rock Mechanics in productivity and protection: 25th U.S. Symposium on Rock Mechanics, Evanston, Illinois, p. 370-380.

PART V: RVT--A FORTRAN PROGRAM FOR AN EXACT ELASTIC SOLUTION FOR TECTONIC AND GRAVITY STRESSES IN ISOLATED SYMMETRIC RIDGES AND VALLEYS

INTRODUCTION

In the RVT[1] FORTRAN program, an exact elastic solution for the stress response of isolated symmetric ridges and valleys to gravitational loading is combined with an exact solution for the stress response of the same features to a far-field uniform tectonic stress. The gravitational-loading solution was originally done by Akhpatelov (Akhpatelov and Ter-Martirosyan, 1971) and has been rederived by Savage and others (in press). The tectonic-loading solution is by Savage (W. Z. Savage and H. S. Swolfs, written commun., 1984). Both solutions are based on the complex potential methods of Muskhelishvili (1953) for plane elasticity and are described in detail in Savage and others (in press) and W. Z. Savage and H. S. Swolfs (written commun., 1984). In this report, we briefly describe the conformal mapping function used to transform isolated symmetric ridges and valleys into a half-plane, give the expressions for stresses for both gravity and tectonic loadings, provide a program listing, describe the input and output variables, and give a sample calculation.

CONFORMAL MAPPING FUNCTION

The conformal transformation for a symmetric ridge into a half-plane is given by

$$z = f(w) = w + \frac{ab}{w - ia} \quad , \tag{1a}$$

or separated into real and imaginary parts by

$$x = u + \frac{abu}{u^2 + (v - a)^2} \tag{1b}$$

$$y = v - \frac{ab(v - a)}{u^2 + (v - a)^2} \quad . \tag{1c}$$

Here, b represents the ridge height. When u = a and v = 0, then x = a + b/2 and y = b/2, which are the coordinates of the inflection point on the ridge flank. The slope at this inflection point is given by dy/dx = -b/2a. These relationships and the conformal transformation are shown in figure 1. The transformation for a symmetric valley is obtained by changing the sign of b in equations (1). Note that for a valley, the parameter a must be greater than the absolute value of b, otherwise a cusp is formed and transformation (1) will no longer be conformal.

---

[1]An abbreviation for ridge-valley-tectonics.



Figure 1.--Conformal mapping given by equation (1) for a symmetric ridge. Relationships between a and b are as shown.

STRESS EXPRESSIONS FOR GRAVITY AND TECTONIC LOADINGS

Gravity-induced stresses in the symmetric ridge or valley defined by equation (1) are obtained from

$$\sigma_x + \sigma_y = -4 \,\mathrm{Re}\, \{\frac{(w - ia)^2}{(w - ia)^2 - ab} [A(w) - \frac{ab\Phi(-ia)}{(w - ia)^2}]\} + \frac{1}{1 - \mu} \rho gy \quad , \tag{2}$$

where Re indicates the real part, ρ is density, g is gravitational acceleration, μ is Poisson's ratio, and where

$$A(w) = - \frac{\rho ga^2 b}{2(1-\mu)}\{\frac{i(4a + b)}{8a^2(w-ia)} + ab[\frac{1}{4a^2(w-ia)^2} - \frac{i}{2a(w-ia)^3}] + (1-2\mu)[\frac{i(4a + b)}{8a^2(w-ia)}\} \quad , \tag{3}$$

and

$$\Phi(-ia) = -\rho gb[\frac{(4a + b)(1 - \mu) + b}{8(1 - \mu)(2a + b)}] \quad , \tag{4}$$

and from the real and imaginary parts of

$$\sigma_y - \sigma_x + 2i\sigma_{xy} = \frac{2(w - ia)^2}{(w - ia)^2 - ab}\{[\frac{\bar{w}(\bar{w} + ia) + ab}{\bar{w} + ia}]\Phi'(w)$$

$$- B(w) - \Phi(w) - [w + \frac{ab}{w + ia}]\Phi'(w) + \frac{ab[\Phi(w) - \Phi(-ia)]}{(w + ia)^2}\} + [\frac{1 - 2\mu}{1 - \mu}]\rho gy \quad , \tag{5}$$

where primes indicate differentiation with respect to w, overbars indicate complex conjugates, and where

$$\Phi(w) = \frac{(w - ia)^2}{(w - ia)^2 - ab} \{ A(w) - \frac{ab\Phi(-ia)}{(w - ia)^2} \} \quad , \tag{6}$$

and

$$B(w) = - \frac{\rho ga^2 b}{2(1-\mu)}\{\frac{i(4a+b)}{8a^2(w-ia)} + (1-2\mu)[\frac{i(4a+b)}{8a^2(w-ia)^2} + ab(\frac{1}{4a^2(w-ia)^2} - \frac{i}{2a(w-ia)^3})]\} \quad . \tag{7}$$

The values of σ_x and σ_y, caused by gravity alone at each point in the ridge or valley, are obtained by forming the sum $\sigma_x + \sigma_y$ given by equation (2) and the difference between $\sigma_y - \sigma_x$ given by the real part of equation (5). The shear stress σ_{xy} is obtained from the imaginary part of equation (5).

Stresses induced by a uniform far-field uniaxial tectonic stress, N_1, which acts normal to the plane of symmetry of the isolated ridge or valley defined by equation (1) are obtained from

$$\sigma_x + \sigma_y = \frac{abN_1[4a + b]}{2a + b} \mathrm{Re}\left[\frac{1}{(w - ia)^2 - ab}\right] + N_1 \quad , \tag{8}$$

and

$$\sigma_y - \sigma_x + 2i\sigma_{xy} = \frac{2(w - ia)^2}{(w-ia)^2-ab} \left\{ \left[\frac{\bar{w}(\bar{w} + ia) + ab}{\bar{w}+ia}\right]\Phi_o'(w) \right.$$

$$\left. + A_o(w) - \Phi_o(w) - \left[w + \frac{ab}{w + ia}\right]\Phi_o'(w) + \frac{ab[\Phi_o(w) - \Phi_o(-ia)]}{(w + ia)^2} \right\} - N_1 \quad , \tag{9}$$

where,

$$\Phi_o(w) = \frac{abN_1}{4} \left[\frac{4a + b}{2a + b}\right] \left[\frac{1}{(w - ia)^2 - ab}\right] \quad , \tag{10}$$

$$A_o(w) = - \frac{abN_1}{2(w - ia)^2} \quad , \tag{11}$$

and

$$\Phi_o(-ia) = - \frac{bN_1}{4(2a + b)} \quad . \tag{12}$$

Again σ_{xy} is obtained from the imaginary part of equation (9) and σ_x and σ_y are obtained by forming sums and differences of $\sigma_x + \sigma_y$ (equation 8) and $\sigma_y - \sigma_x$ from the real part of equation (9).

The two solutions given by equations (2), (5), (8), and (9) are combined in the RVT FORTRAN program. This allows calculation of plane elastic stress fields induced by gravity alone, tectonic loading alone, or a combination of both for topographies which are analogous to those shown in figure 1.

INPUT/OUTPUT VARIABLES

The program was written to run on the VAX 11/780, in VAX-11 FORTRAN (based on American National Standard FORTRAN-77; ANSI X3.9-1978). Extensions to FORTRAN-77 have been avoided. A listing of RVT is given in the appendix. Input variables are, in order, a (distance from center line of ridge or valley to inflection point on the surface), b (maximum ridge height (+) or valley depth (-)), pr (Poisson's ratio), rg (density times the acceleration due to gravity), ts (the far-field tectonic stress; compression (-) a tension (+)), uinc (increment in the u-coordinate), umin (minimum u-coordinate), umax (maximum u-coordinate), vinc (increment in v-coordinate), vmin (minimum v-coordinate), and vmax (maximum v-coordinate; generally taken to be zero). The output variables are, in order, u, v, x, y, sigx, sigy, sigxy, alfa (angle that the algebraically greatest principal stress makes with the x-axis), sigz ($\sigma_z = \mu(\sigma_x + \sigma_y)$; the out-of-plane principal stress), and tmax (the maximum shear stress acting at a point in the body).

INPUT DATA FOR SAMPLE PROBLEM

The following input data are for dimensionless gravity and compressive tectonic stresses in a ridge:

	a	1.00,
	b	1.00,
	pr	0.25,
	rg	1.00,
	ts	-1.00,
	uinc	0.50,
	umin	0.00,
	umax	2.00,
	vinc	0.50,
	vmin	-2.00,
and	vmax	0.00.

All lengths are non-dimensionalized by the value b, and all stresses are non-dimensionalized by ρgb.

PROGRAM OUTPUT PRODUCED USING THE SAMPLE DATA

a= 1.0000 b= 1.0000 pr= 0 2500 rg=0 100E+01 ts=- 100E+01

u	v	x	y	sigx	sigy	sxy	alfa	sigz	tmax
*****	*****	*****	*****	*****	*****	*****	*****	*****	*****
0. 000E+00	-. 200E+01	0 000E+00	- 167E+01	-. 163E+01	-. 218E+01	0 000E+00	0 000E+00	-. 953E+00	0. 2753E+00
0. 500E+00	-. 200E+01	0 554E+00	- 168E+01	- 164E+01	- 218E+01	0. 473E-01	0 493E+01	-. 955E+00	0. 2762E+00
0 100E+01	-. 200E+01	0 110E+01	-. 170E+01	- 166E+01	- 218E+01	0 909E-01	0 956E+01	-. 959E+00	0. 2773E+00
0. 150E+01	-. 200E+01	0. 163E+01	-. 173E+01	-. 168E+01	- 217E+01	0 127E+00	0. 138E+02	-. 964E+00	0 2751E+00
0. 200E+01	-. 200E+01	0. 215E+01	- 177E+01	- 172E+01	-. 215E+01	0 152E+00	0 175E+02	-. 968E+00	0. 2661E+00
0. 000E+00	-. 150E+01	0. 000E+00	- 110E+01	- 145E+01	-. 167E+01	0 000E+00	0. 000E+00	-. 780E+00	0. 1091E+00
0. 500E+00	-. 150E+01	0. 577E+00	-. 112E+01	- 146E+01	- 167E+01	0 607E-01	0. 151E+02	- 784E+00	0. 1209E+00
0. 100E+01	-. 150E+01	0 114E+01	-. 116E+01	- 150E+01	- 168E+01	0. 114E+00	0. 256E+02	-. 794E+00	0. 1460E+00
0. 150E+01	-. 150E+01	0. 168E+01	-. 121E+01	- 154E+01	-. 168E+01	0. 155E+00	0 331E+02	-. 804E+00	0. 1697E+00
0. 200E+01	-. 150E+01	0 220E+01	-. 126E+01	-. 158E+01	- 165E+01	0 180E+00	0. 394E+02	-. 809E+00	0. 1833E+00
0. 000E+00	-. 100E+01	0. 000E+00	- 500E+00	- 124E+01	- 114E+01	0. 000E+00	0. 900E+02	-. 597E+00	0. 5022E-01
0. 500E+00	-. 100E+01	0. 618E+00	-. 529E+00	- 127E+01	- 115E+01	0. 887E-01	0. 620E+02	-. 607E+00	0. 1070E+00
0. 100E+01	-. 100E+01	0. 120E+01	-. 600E+00	- 134E+01	- 117E+01	0. 156E+00	0 589E+02	-. 628E+00	0. 1761E+00
0. 150E+01	-. 100E+01	0. 174E+01	-. 680E+00	-. 141E+01	- 118E+01	0 200E+00	0 600E+02	-. 648E+00	0. 2310E+00
0. 200E+01	-. 100E+01	0 225E+01	-. 750E+00	-. 147E+01	- 115E+01	0 217E+00	0. 634E+02	-. 656E+00	0. 2705E+00
0. 000E+00	-. 500E+00	0. 000E+00	0. 167E+00	-. 968E+00	-. 604E+00	0. 000E+00	0 900E+02	-. 393E+00	0. 1820E+00
0. 500E+00	-. 500E+00	0. 700E+00	0 100E+00	-. 104E+01	-. 626E+00	0. 156E+00	0. 714E+02	-. 416E+00	0 2576E+00
0. 100E+01	- 500E+00	0. 131E+01	- 385E-01	-. 119E+01	-. 674E+00	0 249E+00	0. 681E+02	- 467E+00	0. 3594E+00
0. 150E+01	-. 500E+00	0. 183E+01	-. 167E+00	- 134E+01	-. 685E+00	0 283E+00	0. 695E+02	-. 506E+00	0. 4316E+00
0. 200E+01	-. 500E+00	0 232E+01	-. 260E+00	- 142E+01	-. 632E+00	0. 262E+00	0 732E+02	-. 514E+00	0. 4750E+00
* 0. 000E+00	0. 000E+00	0. 000E+00	0 100E+01	- 556E+00	0 000E+00	0. 000E+00	0. 900E+02	-. 139E+00	0. 2778E+00
0. 500E+00	0. 000E+00	0. 900E+00	0 800E+00	-. 641E+00	-. 120E+00	0 277E+00	0. 666E+02	-. 190E+00	0. 3803E+00
0. 100E+01	0. 000E+00	0. 150E+01	0 500E+00	- 103E+01	-. 258E+00	0 516E+00	0. 634E+02	-. 322E+00	0. 6444E+00
0. 150E+01	0. 000E+00	0. 196E+01	0 308E+00	- 145E+01	- 151E+00	0 468E+00	0 721E+02	-. 401E+00	0. 8019E+00
0. 200E+01	0 000E+00	0 240E+01	0. 200E+00	- 151E+01	- 498E-01	0 274E+00	0 797E+02	-. 389E+00	0 7778E+00

* Crest of the ridge.

Engineering sign convention (tension positive) is used for the stresses. If the variable rg is set equal to 0, only the tectonic effect will be computed for regional compression (-) or regional tension (+). If the variable ts is set equal to zero, only the gravity effect will be computed. A rather broad class of problems may be studied when the gravity body force and the tectonic far-field stresses (positive or negative) are combined, and the shapes of ridges (+) or valleys (-) are varied within permissible limits.

REFERENCES

Akhpatelov, D.M., and Ter-Martirosyan, Z.G., 1971, The stressed state of ponderable semi-infinite domains: Bulletin of Armenian Academy of Sciences, Mechanics, v. 24, no. 3, p. 33-40.

Muskhelishvili, N.I., 1953, Some basic problems of the mathematical theory of elasticity: P. Noordhoff, Ltd., Groningen, The Netherlands, 718 p.

Savage, W.Z., Swolfs, H.S., and Powers, P.S., in press, Gravitational stresses in long symmetric ridges and valleys: International Journal of Rock Mechanics and Mining Sciences and Geomechanics Abstracts.

APPENDIX

Program Listing for RVT.FOR

```
C
C
C     ******************************************************************
C
C                                 RVT.FOR
C
C
C     This program was developed and written by W.Z. Savage and
C      P.S. Powers for the elastic solution of tectonic and gravity
C      stresses in isolated symmetric ridges and valleys.
C
C
C     ******************************************************************
C
      Character*12 file1
      Real a,b,pr,uinc,umin,umax,vinc,vmin,vmax,rad
      Real x,y,u,v,r,so,rg,sum,dif,phia
      Real ts,sumt,phiat,d2phia,dift
      Complex awt,bwt,phit,d1phi1,d1phi2,d1phit
      Complex psi1t,psi2t,psit,str1t
      Complex w,z,po,dz,aw1,aw2,aw,phi,d1aw1,d1aw2,d1aw3
      Complex d1aw,d1phi,d2phi,d2phi2,d2phi3,d2phi1
      Complex bw1,bw2,bw,psi1,psi2,psi,zbar,str1,ai
C
C     ******************************************************************
C
C     Initial Input/Output Section
C
      Write (6,202)
C
C     Read the output file name into the string variable 'file1'.
C
      Read (5,203) file1
      Open (12,name = file1,status = 'new')
C
C
      Write (6,200)
      Read (5,*)a,b,pr,rg,ts,uinc,umin,umax,vinc,vmin,vmax
C
C     The input variables are defined as follows:
C
C     a      distance from center line of ridge or valley to inflection
C            point on the surface,
C     b      maximum ridge height (positive value) or
C            valley depth (negative value),
```

```
C      pr     Poisson's ratio,
C      rg     density times acceleration due to gravity,
C      ts     far-field tectonic stress; compression (negative
C              value), tension (positive value),
C      uinc   increment in the u-coordinate,
C      umin   minimum u-coordinate,
C      umax   maximum u-coordinate,
C      vinc   increment in v-coordinate,
C      vmin   minimum v-coordinate,
C      vmax   maximum v-coordinate.
C
       Write (12,209)a,b,pr,rg,ts
       Write (12,210)
       Write (12,211)
C      rad is the small radius from point w = -ia.
       rad = 1.0e-04
       so = rg*b
       po = cmplx(0.0,1.0)
       ai = po*a
C      ************************************************************
C
C      Phi at w = -ia for gravity solution from equation 4.
C
       phia = -so*((4.*a+b)*(1.-pr)+b)/(8.*(1.-pr)*(2.*a+b))
C
C      ************************************************************
C
C      phiat is phi at w = -ia for tectonic solution from equation 12.
       phiat = -(ts*b)/(4.*(2.*a+b))
C      d2phia is the 2nd derivative of phi at w = -ia for tectonic sol.
       d2phia = -(ts*b*(4.*a+b)*(b-12.*a))
       d2phia = d2phia/(2.*a*(2.*a+b)*(4.*a+b)**3)
       numax = int((umax-umin)/uinc)
       nvmax = int((vmax-vmin)/vinc)
       n = 0
       m = 0
       v = vmin
   100 u = umin
   150 w = cmplx(u,v)
       r = sqrt(u**2+(v+a)**2)
C      ************************************************************
C
C      From equation 1.
C
       z = w+(a*b)/(w-ai)
C
C      ************************************************************
       x = Real (z)
       y = aimag(z)
       dz = ((w-ai)**2-a*b)/((w-ai)**2)
C      ************************************************************
C
```

```
C      aw is a(w) given by equation 3 in text.
C
       aw1 = po*(4.*a+b)/(8.*(w-ai))
       aw2 = a*b*(w-3.*ai)/(8.*(1.-pr)*(w-ai)**3)
       aw = -aw1-aw2
C
C      ************************************************************
C
C      phi is phi(w) for the gravity solution given by equation 6.
C
       phi = -aw*so/dz+a*b*phia/(dz*(w-ai)**2)
C
C      ************************************************************
C
C      From equation 11.
C
       awt = -(a*b*ts)/(2.*(w-ai)**2)
C
C      ************************************************************
C
C      From equation 10.
C
       phit = -awt/dz+a*b*phiat/(dz*(w-ai)**2)
C
C
C      ************************************************************
C
C      From equation 8.
C
       sumt = 4.*Real (phit)+ts
       sum = sumt+4.*Real (phi)+rg*y/(1.-pr)
C
C      ************************************************************
C
C      First derivative of phi(w) in equation 5.
C
       d1aw1 = po*(4.*a+b)/(8.*(w-ai)**2)
       d1aw2 = 2.*a*b/(8.*(1.-pr)*(w-ai)**3)
       d1aw3 = 6.*po*a**2*b/(8.*(1.-pr)*(w-ai)**4)
       d1aw = d1aw1+d1aw2-d1aw3
       d1phi = -so*d1aw/dz-(2.*a*b*(phi+phia))/(dz*((w-ai)**3))
C
C      ************************************************************
C
C      Second derivative of phi(w) to be used in equation 5
C      when w = -ia
C
       d2phi1 = 2.*phi/((w-ai)**2)
       d2phi2 = 4.*d1phi/(w-ai)
       d2phi3 = so*po*a**2*b/(2.*(1.-pr)*((w-ai)**5))
       d2phi = -(d2phi1+d2phi2+d2phi3)/dz
C
```

```
C     ***********************************************************
C
C     From equation 7.
C
      bw1 = po*(4.*a+b)/(8.*(w-ai))
      bw2 = (1.-2.*pr)*a*b*(w-3.*ai)/(8.*(1.-pr)*(w-ai)**3)
      bw = -so*(bw1+bw2)
C
C     ***********************************************************
      psi1 = w*d1phi+bw+phi
      d1phi1 = -(a*b*ts*(4.*a+b)*(w-ai))
      d1phi2 = 2.*(2.*a+b)*(((w-ai)**2-a*b)**2)
      d1phit = d1phi1/d1phi2
      bwt = -awt
C     ***********************************************************
C
C     Part of equation 9.
C
      psi1t = w*d1phit+bwt+phit
C
C     ***********************************************************
C
C     Test on closeness of w to -ia.  If w is near -ia, the Taylor
C     expansion about -ia is used
C
      If (r.LT.rad)Go To 160
C
C     ***********************************************************
C
C
      psi2 = a*b*d1phi/(w+ai)-a*b*(phi-phia)/((w+ai)**2)
      psi2t = a*b*d1phit/(w+ai)-a*b*(phit-phiat)/((w+ai)**2)
      Go To 170
C     ***********************************************************
C
C     Taylor's expansion for gravity and tectonic stress at w = ia.
C
  160 psi2 = .5*a*b*d2phi
      psi2t = .5*a*b*d2phia
C
C     ***********************************************************
  170 psi = -(psi1+psi2)/dz
      psit = -(psi1t+psi2t)/dz
      zbar = conjg(z)
C     ***********************************************************
C
C     From equation 5.
C
      str1 = 2.*(zbar*d1phi/dz+psi)
C
C     ***********************************************************
C
```

```
C     From equation 9.
C
C
C     ***************************************************************
C
C     The equations in this block form the sums and differences
C     of equations 2, 5, 8, and 9, to obtain stresses.
C
      dift = Real (str1t)-ts
      dif = dift+Real (str1)+rg*y*(1.-2.*pr)/(1.-pr)
      sigxy = aimag(str1)+aimag(str1t)
      sigxy = sigxy/2.
      sigx = (sum-dif)/2.
      sigy = (sum+dif)/2.
C
C     ***************************************************************
C
C     Principal stress directions.
C
      alfa = .5*atan2((2.*sigxy),(sigx-sigy))
      alfa = (180./3.14159)*alfa
C
C     ***************************************************************
C
C     Principal stess magnitudes.
C
      sig1 = sum/2.+sqrt((dif/2.)**2+sigxy**2)
      sig2 = sum/2.-sqrt((dif/2.)**2+sigxy**2)
C
C     ***************************************************************
C
C     Maximum shear stress.
C
      tmax = (sig1-sig2)/2.
C
C     ***************************************************************
C
C     Out of plane stress.
C
      sigz = pr*(sigx+sigy)
C
C     ***************************************************************
      Write (12,275)u,v,x,y,sigx,sigy,sigxy,alfa,sigz,tmax
      n = n+1
      u = u+uinc
      If (n.LE.numax)Go To 150
      n = 0
      m = m+1
      v = v+vinc
      If (m.LE.nvmax)Go To 100
C
C     **************************************************************************
```

```
C
C     Format statements
C
  200 Format (2x,'Enter a,b,pr,rg,ts,uinc,umin,umax,vinc,vmin,vmax')
  202 Format (2x,'$   Enter the name of the output file.')
  203 Format (a)
  209 Format (5x,'a=',f12.4,5x,'b=',f12.4,5x,'pr=',f12.4
     &,5x,'rg=',e9.3,5x,'ts=',e9.3)
  210 Format (x,t5,'u',t16,'v',t27,'x',t40,'y',t52,'sigx',t65,'sigy',
     &t77,'sxy',t90,'alfa',t103,'sigz',t117,'tmax')
  211 Format (x,t4,'*****',t14,'*****',t24,'*****',t38,'*****',t50,
     &'*****',t63,'*****',t75,'*****',t88,'*****',t101,'*****',t115,
     &'*****')
  275 Format (x,e9.3,2x,e9.3,2x,e9.3,2x,e9.3,4x,e9.3,4x,
     &e9.3,4x,e9.3,4x,e9.3,4x,e9.3,4x,e10.4)
      Stop
      End
```